Perlの「関数」は、与えられた値に基づいて定義された独自の処理を実行し、その結果を返す一塊のコードのことです。
英語では関数のことを ''function'' と呼び、「機能」と訳すことができます。
Perlの「関数」には、言語コアで定義済みの「組込み関数」とユーザーが定義する「サブルーチン」とがあります。
サブルーチンをつくる場合にも、最終的には、プログラマーの手により「組込み関数」「式」や「文」をくみあわせて作成することになります。
前節で紹介した print 関数は、組込み関数です。
このように、組込み関数は、名前を使って呼出すだけで使えます。
いっぽう、サブルーチンは、名前を使って呼び出す以前に、処理の内容もプログラマーが作成する（サブルーチを定義する）必要があります。
Perlの言語コアで定義済みの関数のことを「組込み関数」と言います。
中には my, use や do の様に、見るからに構文要素なものも「組込み関数」に分類されています。
機能:print関数は、引数で与えられた文字列や文字列のリストを標準出力に出力します。引数が与えられなかったときは $_ が出力されます。
[https://paiza.io/projects/Isiw6BqruLhVNGQFCFCSQQ?language=perl 例]:
use v5.30.0;
use warnings;
print "Hello, World\n";
print "Hello, Perl\n"
実行結果:
Hello, World 
Hello, Perl
print関数は、行末で改行しないので、もし改行をしたい場合には明示的にエスケープシーケンス \n を文字列の末尾に加えます。
Perl 5.10 から導入されたsay 関数は、行末で改行を行います。これで、都度 \n を文字列末に記述する手間が省けます。
[https://paiza.io/projects/kCCvuvfgVSM9IGX84QE1cg?language=perl 組込み関数 say]:
use strict;
use warnings;
use utf8;
binmode STDOUT,":encoding(UTF-8)";
use feature "say";
use feature ':5.10';
use v5.10;
say "Hello";
say "Yes!";
say "Boodbye";
my $message = "こんにちは";
say $message;
say を使うには、6-8 行目の use 宣言のいずれか１つが必要です。
use feature "say";
use feature ':5.10';
use v5.10;
use feature "say";
say を名指しで有効化しています。お勧めです。
use feature qw(say switch);
の様に２つ以上を列挙することもできます。
use feature ':5.10';
バージョン 5.10 以降の機能を全て有効にします。手早く動かすためにはいいのですが過剰です。
use v5.10;
意味的には上と同じですが、より簡素です。多分一番多く使われています。
CORE::say:
use strict;
use warnings;
CORE::say "Hello world!";
CORE::を前置するとプラグマを使わずに say関数を使うことができます。
ワンライナーや書き捨てのスクリプトに向いています。
CORE はPerlコアルーチンの名前空間です。
Perlでは、文字列の中に変数や式を埋め込むことができ、テンプレート言語であるかのような使いかたが出来ます。
length は文字列の長さを返します。
[https://paiza.io/projects/dTI0m8Arb3XFoZUXcGo4kA?language=perl 文字列に変数や式を埋込む]:
use v5.30.0;
use warnings;
my $x = "aeiou";
my $tmp = length $x;
say "length \"$x\" -> $tmp";
say "length \"aeiou\" -> @{[length 'aeiou']}";
say qq(length "aeiou" -> @{[length 'aeiou']});
実行結果:
length "aeiou" -> 5
length "aeiou" -> 5 
length "aeiou" -> 5 
この様に、”（ダブルクォーテーションマーク）に囲まれた文字列の中では $変数 で式の値が、@{[式]} で式の値が文字列に埋込まれます。
厳密に解説するには、スカラーコンテキストとリストコンテキストの説明が必要なのですが、リファレンスなどの説明が必須なので、機会を見て一括して解説します。
”（ダブルクォーテーションマーク）ではなく、’（シングルクォーテーションマーク）で囲まれた文字列では、変数や式は展開されません。
平方根などの数学計算をする関数が用意されています。
[https://paiza.io/projects/AjzHuhdjMmEhQSMbw_Ufsg?language=perl 最小のピタゴラス数]:
use v5.20.0;
use warnings;
say "sqrt(3**2 + 4**2) --> @{[sqrt(3**2 + 4**2)]}";
use POSIX "hypot";
say "hypot(3, 4) --> @{[ hypot(3, 4) ]}"
実行結果:
sqrt(3**2 + 4**2) --> 5 
hypot(3, 4) --> 5
Perlの組込み関数 sqrt を使って自乗和の平方根を求めています。
自乗は結果がオーバーフローあるいはアンダーフローを起こす可能性があるので、対策された hypot を使うのが定石です。
ですが、Perlの組込み関数にもMathモジュールにも hypot はなく、POSIXモジュールにあります。
この場合、use POSIX "hypot";ではなくuse POSIX;で充分なのですが、POSIXからhypotを持ってきている意外性を伝えるため明示しました。
呼出し側で、POSIX::hypot(3, 4) とするのも刺激的ですが、複数箇所あると鬱陶しいので use 側で対処しました。
桁あふれ対策と可変引数に対応したPerl版hypotの例。
[https://paiza.io/projects/kwxdBFQhFJD82gYMtbA0Vw?language=perl hypot.pl]:
use v5.30.0;
use warnings;
use POSIX;
sub hypot {
    my ( $max, $s ) = ( 0, 0 );
    foreach my $n (@_) {
        next      if $n == 0;
        return $n if $n != $n;       # for NaN
        my $arg = abs($n);
        return $n if $n == "Inf";    # for Inf
        if ( $arg > $max ) {
            $s *= ( $max / $arg ) * ( $max / $arg ) if $max != 0;
            $max = $arg;
        }
        $s += ( $arg / $max ) * ( $arg / $max );
    }
    return $max * sqrt($s);
}
if ( $0 eq __FILE__ ) {
    foreach my $i ( -1075 .. -1073, -540 .. -538, 0 .. 2, 508 .. 511, 1021 .. 1024 ) {
        my $j = 2**$i;
        my ( $n, $m ) = ( 3 * $j, 4 * $j );
        say "$i: @{[ 5 * $j ]} @{[ sqrt($n*$n + $m*$m) ]} @{[ ::hypot($n, $m) ]} @{[ POSIX::hypot($n, $m) ]}";
    }
}
実行結果:
-1075: 0 0 0 0
-1074: 2.47032822920623e-323 0 2.47032822920623e-323 2.47032822920623e-323
-1073: 4.94065645841247e-323 0 4.94065645841247e-323 4.94065645841247e-323
-540: 1.38922421842817e-162 0 1.38922421842817e-162 1.38922421842817e-162
-539: 2.77844843685635e-162 3.14345556940526e-162 2.77844843685635e-162 2.77844843685635e-162
-538: 5.55689687371269e-162 5.44462475754526e-162 5.55689687371269e-162 5.55689687371269e-162
0: 5 5 5 5
1: 10 10 10 10
2: 20 20 20 20
508: 4.18993997810706e+153 4.18993997810706e+153 4.18993997810706e+153 4.18993997810706e+153
509: 8.37987995621412e+153 8.37987995621412e+153 8.37987995621412e+153 8.37987995621412e+153
510: 1.67597599124282e+154 Inf 1.67597599124282e+154 1.67597599124282e+154
511: 3.35195198248565e+154 Inf 3.35195198248565e+154 3.35195198248565e+154
1021: 1.12355820928895e+308 Inf 1.12355820928895e+308 1.12355820928895e+308
1022: Inf Inf Inf Inf
1023: Inf Inf Inf Inf
1024: Inf Inf Inf Inf 
Perlには、Cの isnan() や isfinite() に相当する関数がないので、それぞれ $n != $n と abs($n) == "Inf" としました。
POSIXモジュールにはisfinite関数があるので、それを使えばよいのですが、POSIX::hypotの代替実装なので利用を見送りました。
sin,cos は組込み関数にありますが、tan, acos など他の三角関数や円周率（pi）を使用するには、use宣言を使って Math::Trigモジュールから導入します。
[https://paiza.io/projects/GyO8xRXjKsGkXzuvl2L9wQ?language=perl 余弦関数と逆余弦関数]:
use 5.30.0;
use warnings;
use Math::Trig qw(pi acos);
say "cos(pi) -> cos(@{[pi]}) -> @{[cos(pi)]}";
say "acos(-1) ->  @{[acos(-1)]}"
実行結果:
cos(pi) -> cos(3.14159265358979) -> -1 
acos(-1) ->  3.14159265358979
円周率は、Math::Trigモジュールを導入すると使えるようになりますが、$piではなく piです。
文字列中で参照する場合は "@{[pi]}" となります。
Perlの三角関数の角度の単位は多くのプログラミング言語同様ラジアン（弧度法）です。
正弦sinと余弦cosはPerlの言語コアで定義されていますが、正接tanはMath::Trigモジュールで定義されています。
Math::Trigモジュールでは、piなどの定数や他の三角関数関連の諸関数が定義されています。
現在の日時や時刻などを表すには、time関数およびlocaltime関数を使います。
[https://paiza.io/projects/4-FKQ--QuCFTcHN321xcsA?language=perl エポックからの秒数と、ローカル時刻]:
use v5.30;
use warnings;
use utf8;
binmode STDOUT,":encoding(UTF-8)";
my ($sec,$min,$hour,$mday,$mon,$year,$wday,$yday,$isdst) = localtime(time());  
say "time() -> @{[time()]}";
say "いまは、@{[1900 + $year]} 年 @{[1 + $mon]} 月 $mday 日 $hour 時 $min 分 $sec 秒です。";
use POSIX "strftime";
say strftime "%Y/%m/%d %H:%M:%S", localtime();
say strftime "%c", localtime();
実行結果:
time() -> 1668851859
いまは、2022 年 11 月 19 日 9 時 57 分 39 秒です。
2022/11/19 09:57:39 
Sat 19 Nov 2022 09:57:39 AM UTC
説明
time関数は、エポック（1970年1月1日0時0分0秒(UTC) ）からの通算秒を返します。
localtime関数は、エポックからの通算秒形式の引数を、年月日時分秒の要素に分解しリストで返します。
localtime関数は、引数を省略すると time()が仮定されるので、この例での引数は冗長です。
localtimeが返すリストを操作するには1900を足したり月数の補正をしたり面倒です（よく間違えます）。
POSIXモジュールの strftime を使うと、Cのstrftime()と同じ（正確にはPOSIXと同じ）書式化文字列がつかえ可読性も向上します。使いましょう。
DateTimeモジュールもあるのですが、Perl流のオブジェクト指向の構文で書かれているので、直感的とは言い難いコードになります。使うなとまでは言いません。
split関数には、与えられたパターンで文字列を区切り、リストで返します。
[https://paiza.io/projects/8JO5ecVl3w4IKNg0tncUmg?language=perl split]:
use v5.30;
use warnings;
use utf8;
binmode STDOUT,":encoding(UTF-8)";
my @list = split(/ /, '睦月 如月 弥生 卯月 皐月 水無月 文月 葉月 長月 神無月 霧月 師走');
for (my $i = 0; $i <= $#list; $i++){
  say qq(@{[$i+1]}月: $list[$i]);
}
実行結果:
1月: 睦月
2月: 如月
3月: 弥生
4月: 卯月
5月: 皐月
6月: 水無月
7月: 文月
8月: 葉月
9月: 長月
10月: 神無月
11月: 霧月 
12月: 師走
Perlでは、ユーザーが定義する関数のことをサブルーチン( ''subroutine'' )と呼び、キーワードsubを使い定義します。
サブルーチンの定義と呼出しは、説明することがほとんどないほど簡単です。
[https://paiza.io/projects/ahyDWYUS_WdFUG1H1TAKbw?language=perl シンプルなサブルーチンの定義と呼出し]:
use v5.30.0;
use warnings;
sub world {
    say "Hello, World";
}
sub perl {
    say "Hello, Perl";
}
&world
&perl
world;
perl;
実行結果:
Hello, World
Hello, Perl
見たままですが、&が前置されていなくても、されていても同じというのは釈然としません。
この & は、組込み関数では前置できません。
というわけで、& を関数呼出しで前置するのは、「組込み関数ではなくサブルーチンを呼んでいます」という意味になります。
また、& を省略するとサブルーチンの宣言より前に、サブルーチンを呼出すことはできません。
サブルーチンの定義より先にサブルーチンを呼出す必要があることがあります（典型的には、お互いに呼び合う関数）。
この場合は、呼出ごとに & を前置するか、サブルーチン宣言をサブルーチン呼出の前にします。
[https://paiza.io/projects/YCbleN8uFaa1BNzXrHB9mQ?language=perl サブルーチン宣言]:
use v5.30.0;
use warnings;
&world
&perl
sub world;
world;
sub perl;
perl;
sub world {
    say "Hello, World";
}
sub perl {
    say "Hello, Perl";
}
実行結果:
Hello, World
Hello, Perl
Hello, World 
Hello, Perl
4,5は & を前置しているので、宣言がなくてもサブルーチンとわかる。
7,10がサブルーチン宣言で、サブルーチン定義の前方参照を解決します。
前出の例は、ほとんど同じ内容のサブルーチンを２つ用意しましたが、１つにまとめてみましょう。
[https://paiza.io/projects/MBw8XjvoxM7yMMXkbd1CdQ?language=perl グローバル変数を使ったサブルーチンの振る舞いの変更]:
use v5.30.0;
no strict;
use warnings;
$who = "WHO!";
sub hello {
    say "Hello, $who";
}
&hello
$who = "world";
&hello
$who = "Perl";
&hello
実行結果:
Hello, WHO!
Hello, world 
Hello, Perl
グローバル変数 $who を使ってメッセージの後半を変えています。
この方法はこの方法で動きますし、かつてのFORTRANやBASICは、まさにこのようにグローバル変数でサブルーチンをコントロールしていました。
しかし、２行目のno strict;で明示的に strict を無効にしなければエラーが出るほど、グローバル変数の使用は推奨されない方法です。
前出の例は、グローバル変数を使っていましたが、グローバル変数はデーターフロー的なスパゲティーコードに直結するので、引数を使ってスマートに実装してみましょう。
[https://paiza.io/projects/NvuWuRBObB3oqD2kq1iFnQ?language=perl 引数を使ったサブルーチンの振る舞いの変更]:
use v5.30.0;
use warnings;
sub hello {
    my $who = shift;
    $who //= "WHO?!";
    say "Hello, $who";
}
&hello();
&hello("world");
&hello("Perl");
実行結果:
Hello, WHO?!
Hello, world 
Hello, Perl
my $who = shift;
Perlのサブルーチンの引数は、名前を持った仮引数ではなく特殊変数 @_ に配列として渡されます。
第一引数が $_[0] となります。
半ば定型文なのですが、引数を左から順に読む動作が shift（shift @_ の意味） と符合するので、それらしい名前（この場合は $who）の変数を宣言し shift で初期化するコードが良く見られます。
キーワード my を前置して宣言した変数は、「レキシカル変数」となり、サブルーチン（この場合は hello）を抜けると参照できなくなり、もう一度同じサブルーチンを呼んでも、もう違う値になってます（非永続的なレキシカルスコープ）。
$who //= "WHO?!";
Perlには、ディフォルト引数の構文はなかったので、引数が渡されなかった場合の既定値（ディフォルト）を指定するには、このようなイディオムになります。
この @_ による引数の受渡しは、Perlでは約２０年に渡って使われてきましたが、他のプログラミング言語のように名前付きの仮引数が欲しいとの要望は根強く、シグネチャーとしてv5.20.0から実験的な機能として実装されています。
ここまでで、引数を受取りサブルーチンの振舞いを変えることができるようになりました。
次に、「値を返す手段」が問題になります。
グローバル変数を使って値を返せそうですが「データーフロー的なスパゲティーコード」になるのでサブルーチンの「戻値」を使ってみましょう。
いままでのサブルーチンは値を返しませんでしたが、Perlのサブルーチンは値を１つ返すことができます。
[https://paiza.io/projects/NBXcY50TNs_cU1xGQcEOmg?language=perl 戻値を返すサブルーチン]:
use strict;
use warnings;
sub add {
    my ($x, $y) = @_;
    return $x + $y;
}
print("add(12, 9) -> @{[add(12, 9)]}\n");
print("add(1.2, 0.9) -> @{[add(1.2, 0.9)]}\n");
print("add(123, '89') -> @{[add(123, '89')]}\n");
実行結果:
add(12, 9) -> 21
add(1.2, 0.9) -> 2.1 
add(123, '89') -> 212
return $x + $y;
return 式 ;
での「式」の値が返ります。
return $x + $y;
$x + $y;
と同じです。
Perl の ; は、Cのように式を文にするのではなく、式と式を区切るデリミターなので最後の式の後に  ; は不要です。
print("add(123, '89') -> @{[add(123, '89')]}\n");
add(123, '89') -> 212
となるように、数値が期待されす文脈に数値に変換できる文字列が来ると、自動的に数値に変換され演算されます。
Perlのこの暗黙の変換は、エポックからの通算秒が桁上りしたときなどに発現する凶悪なバグの原因になってきました。
引数と戻値が手に入ったので、再帰的呼出しを行うサブルーチンを書いてみます。
整数の累乗を返すサブルーチン pow を書いてみました。
[https://paiza.io/projects/it2hilVOZ9TGu_lmM68elw?language=perl 整数の冪乗]:
use v5.30.0;
use warnings;
sub pow {
    my ($n, $m) = @_;
    return "Domain error" if $m < 0;
    return 1 if $m == 0;
    return $n if $m == 1;
    return $n * &pow($n, $m - 1);
}
say "pow(2, $_) -> @{[ pow(2, $_) ]}" foreach -1..3;
実行結果:
pow(2, -1) -> Domain error
pow(2, 0) -> 1
pow(2, 1) -> 2
pow(2, 2) -> 4 
pow(2, 3) -> 8
9 で、pow自身を指数を１減らして呼んでいます。
$n, $m が重複使用されているように見えますが、myがついているので、再帰レベルが１つ下るごとに別のインスタンスが生成されています。
整数を3桁ごとにカンマで区切って表示するサブルーチン comma3 を書いてみました。
[https://paiza.io/projects/urtCz3eFlaG7tRvYLm1Qfw?language=perl 整数を3桁ごとにカンマで区切って表示する]:
use v5.30.0;
use warnings;
sub comma3 {
    my $n = shift;
    return "-" . comma3(-$n) if $n < 0;
    my ($num, $rem) = (int($n / 1000), $n % 1000);
    return comma3($num) . sprintf ",%3d", $rem if $num;
    return sprintf "%d", $rem
}
say comma3 $_ foreach qw(
    123456789 
    -999999 
    0 
    1 
    12 
    123 
    1234
)
実行結果:
123,456,789
-999,999
0
1
12
123
1,234
再帰で必ず取上げられるフィボナッチ数列とメモ化を題材に、ベンチマークテストを行ってみようと思います。
[https://paiza.io/projects/tndgcBdq34woengCngWc0w?language=perl フィボナッチ数列とメモ化とベンチマーク]:
use v5.30.0;
use warnings;
sub fibonacci {
    my $n = shift;
    return $n if $n == 0;
    return $n if $n == 1;
    return fibonacci($n - 2) + fibonacci($n - 1)
}
sub fibonacci_norec {
    my $n = shift;
    my ($x, $y) = (1, 0);
    ($x, $y) = ($y, $x + $y) foreach 1..$n;
    return $y
}
sub fibonacci_memorization {
    my $n = shift;
    state @table = (0, 1);
    return $table[$n] if defined $table[$n]; 
    return $table[$n] = fibonacci($n - 2) + fibonacci($n - 1)
}
use Benchmark qw/timethese cmpthese/;
my $i = 16;
cmpthese timethese(2 ** 10, {
  "再帰"   => sub { fibonacci($_) foreach 1..$i },
  "非再帰" => sub { fibonacci_norec($_) foreach 1..$i },
  "メモ化" => sub { fibonacci_memorization($_) foreach 1..$i },
});
実行結果:
Benchmark: timing 1024 iterations of メモ化, 再帰, 非再帰...
 メモ化:  0 wallclock secs ( 0.00 usr +  0.00 sys =  0.00 CPU)
            (warning: too few iterations for a reliable count)
    再帰:  2 wallclock secs ( 1.58 usr +  0.00 sys =  1.58 CPU) @ 648.10/s (n=1024)
 非再帰:  0 wallclock secs ( 0.01 usr +  0.00 sys =  0.01 CPU) @ 102400.00/s (n=1024)
            (warning: too few iterations for a reliable count)
                           Rate              再帰         非再帰  メモ化
再帰                    648/s                  --              -99%      -100%
非再帰              102400/s              15700%                --      -100% 
メモ化 1023999999999999872/s 157999999999999968% 1000000000000001%         --
fibonacci は、素朴な再帰版のフィボナッチ数列です。
fibonacci_norec は、非再帰版のフィボナッチ数列です。
fibonacci_memorization は、メモ化を施した再帰版のフィボナッチ数列です。
20行目の state @table = (0, 1);は、非揮発性のレキシカルスコープ変数の宣言で、 my と違い最初しか初期化されず、再び同じサブルーチンを呼ばれたときには前の値を憶えています。またサブルーチンの外から参照する方法はありません。
メモ化は、一度計算した答えを記憶して次からは記憶から答える戦略なので、ベンチマークに有利です。
メモ化を行うアルゴリズムと行なわないアルゴリズムでは、ベンチマークのような繰り返しに関する感受性が違います。繰返し回数に対し線形に時間が増えないアルゴリズムはメモ化を行っている可能性があるので、ループの底で使われるのには適していますが、頻度の低い使い方の場合、性能が予想より悪い可能性があります。
このことから、実際のプログラムのプロファイル結果とベンチマークの結果の傾向の比較も重要になります。
サブルーチン自身へのリファレンス __SUB__ を使うと無名関数の再帰ができます。
[https://paiza.io/projects/it2hilVOZ9TGu_lmM68elw?language=perl 整数の冪乗]:
use v5.30.0;
say (sub {
    my $n = shift;
    return $n == 0 ? $n
         : $n == 1 ? $n
         : $n * __SUB__->( $n - 1 );
}->(7));
実行結果:
5040
前出の例で、引数を使ってサブルーチンの振る舞いを変えることができました。
機能的には充足しているのですが、名前付きの仮引数を望む声は以前からあり、Perl 5.20.0 から「実験的」( ''experimental'' )なシグネチャーの実装が行なわれ、Perl 5.26.0 からコア機能の１つとなりました。
[https://paiza.io/projects/9iJ8toU2PS78IlvlqtcFow?language=perl シグネチャー]:
use v5.30;
use feature 'signatures';
no warnings "experimental::signatures";
 
sub hello($who = "WHO?!") {
    say "Hello, $who";
}
&hello();
&hello("world");
&hello("Perl");
実行結果:
Hello, WHO?!
Hello, world 
Hello, Perl
sub hello($who = "WHO?!") {
名前を持った仮引数が使えるようになりました。
ディフォルト引数にも対応しています。
use v5.30;
use feature 'signatures';
no warnings "experimental::signatures";
use v5.36.0;
とできます（使用している処理系が、v5.30.0以降の場合に限ります）。
sort のように、コードブロックを引数とするサブルーチンを考えてみましょう。
[https://paiza.io/projects/OtxUGCrI2TXyqXxZsWWHDw?language=perl 例]:
use v5.30.0;
sub bis(&) {
    my $cbr = shift;
    $cbr->();
    $cbr->()
}
bis { say 'Hello, world!' };
my $i = 0;
bis { $i++ };
say $i;
実行結果:
Hello, world!
Hello, world! 
2
与えられたコードブロックを２回実行するサブルーチンです。
3行目のsub bis(&)の&はラムダ抽象です。
組込み関数 map を模倣したサブルーチン mapx を実装します。
[https://paiza.io/projects/GWnq66NY0nfz5jDrvjP3pQ?language=perl 例]:
use v5.30.0;
sub map(&@) {
    my ( $cbr, @ary ) = @_;
    my @result;
    push @result, $cbr->( local $a = $_ ) foreach @ary;
    return @result;
}
say main::map { 2 * $_ } ( 1, 2, 3 );
say main::map { 2 * $a } ( 1, 2, 3 );
say CORE::map { 2 * $_ } ( 1, 2, 3 );
say main::map { $_ x 2 } qw(a b c);
say main::map { $a x 2 } qw(a b c);
say CORE::map { $_ x 2 } qw(a b c);
実行結果:
246
246
246
aabbcc
aabbcc 
aabbcc
組込み関数 sort の様に、$a でコードブロックに引数を渡すこともできるようにしました。
local で宣言しているので、スコープは foreach 式の中だけで、抜けるとグローバルな $a は埋戻されます。
組込み関数に reduce がなかったので実装しました。
[https://paiza.io/projects/Oow9-sgUyfHJ2HcPcyoH2w?language=perl 例]:
use v5.30.0;
use warnings;
sub reduce(&@) {
    my ( $cbr, @ary ) = @_;
    my $init = shift @ary;
    $init = $cbr->( local $a = $init, local $b = $_ ) foreach @ary;
    return $init;
}
say reduce { $_[0] + $_[1] } 1 .. 10;
say reduce { $_[0] . $_[1] } "A" .. "Z";
say reduce { $a + $b } 1 .. 10;
say reduce { $a . $b } "A" .. "Z";
実行結果:
55 
ABCDEFGHIJKLMNOPQRSTUVWXYZ
55 
ABCDEFGHIJKLMNOPQRSTUVWXYZ
組込み関数 sort の様に、$a と $b でコードブロックに引数を渡すこともできるようにしました。
組込み関数に grep は、多くの言語で filter の名前で知られる関数です。
[https://paiza.io/projects/Zm4tMKt5pp0--A61grEvGg?language=perl 例]:
use v5.30.0;
sub filter(&@) {
    my ( $cbr, @ary ) = @_;
    my @result = ();
    $cbr->( local $a = $_ ) ? push( @result, $_ ) : 0 foreach @ary;
    return @result;
}
say filter { $_ % 2 == 1; } 1 .. 10;
say filter { $a % 2 == 1; } 1 .. 10;
say grep { $_ % 2 == 1; } 1 .. 10;
say filter { index( "Hello world", $_ ) >= 0 } ( "A" .. "Z", "a" .. "z" );
say filter { index( "Hello world", $a ) >= 0 } ( "A" .. "Z", "a" .. "z" );
say grep { index( "Hello world", $_ ) >= 0 } ( "A" .. "Z", "a" .. "z" );
say "@{[ map { $_ * 2 } filter { $_ % 2 == 1; } 1..10]}";
say "@{[ map { $_ * 2 } filter { $a % 2 == 1; } 1..10]}";
say "@{[ map { $_ * 2 } grep { $_ % 2 == 1; } 1..10]}";
実行結果:
13579
13579
13579
Hdelorw
Hdelorw
Hdelorw
2 6 10 14 18
2 6 10 14 18 
2 6 10 14 18
組込み関数 sort の様に、$a でコードブロックに引数を渡すこともできるようにしました。
[TODO:スコープルールに関する簡素な説明]
[TODO:コンテキストに関する例をふんだんに使った解説]
my で宣言した変数（レキシカル変数）はサブルーチンを抜けると御破算になりますが、state で宣言した変数はレキシカルスコープであるものの次にサブルーチンが呼ばれたときも値を憶えています。
[https://paiza.io/projects/BUEgIyXqdobwwm9p3VsmKw?language=perl state $var]:
use v5.10;
sub func {
    my $myVar = 0;
    state $stateVar = 0;
    $myVar++;
    $stateVar++;
    CORE::say "\$myVar = $myVar, \$stateVar = $stateVar";    
}
&func for 1..5
実行結果:
$myVar = 1, $stateVar = 1
$myVar = 1, $stateVar = 2
$myVar = 1, $stateVar = 3
$myVar = 1, $stateVar = 4
$myVar = 1, $stateVar = 5
state $var は、Perl 5.10 以降でサポートされています。
同じサブルーチンを呼出しても、スカラーが戻ることを期待している文脈（スカラーコンテキスト）と、リストが戻ることを期待している文脈（リストコンテキスト）の２通りがあります。
この２つのケースを判別するために wantarray 関数が用意されています。
[https://paiza.io/projects/wjGlO4nNueGJHx_A17Zh-g?language=perl コンテキストとwantarray関数]:
use strict;
use warnings;
sub func {
    return map { $_ * 2 } @_ if wantarray();
    my $sum = 0;
    $sum += $_ for @_;
    return $sum;
}
my @x = func(0, 1, 2);
my $y = func(0, 1, 2);
print < @x
my \$y = func(0, 1, 2); \$y -> $y
@{[ func(1,2,3) ]}
@{[ scalar func(1,2,3)]}
@{[ ~~ func(1,2,3)]}
EOS
実行結果:
my @x = func(0, 1, 2); @x -> 0 2 4
my $y = func(0, 1, 2); $y -> 3
2 4 6
6 
6
wantarray 関数は、サブルーチンを呼出したコンテキストで戻値としてリストが要求されているなら真を、スカラーが要求されているなら偽を返します。
関数 func は、リストコンテキストでは全ての要素を二倍にしたリストを、スカラーコンテキストでは全ての要素の合計を返します。
chomp chop chr crypt fc hex index lc lcfirst length oct ord pack q/STRING/ qq/STRING/ reverse rindex sprintf substr tr/// uc ucfirst y///
書式:index STR, SUBSTR [, POSITION]
機能:文字列STRの中で、部分文字列SUBSTRが最初に出現する位置を返します。
[https://paiza.io/projects/PYUKqfDhk7sSi9l1Sdwsxg?language=perl 組込み関数 index]:
use v5.30.0;
use warnings;
my $str = "This is a pen.";
my $substr = "pen";
say qq(index "$str", "$substr" -> @{[index $str, $substr]});
$str = "これは、ペンです。";
$substr = "ペン";
say qq(index "$str", "$substr" -> @{[index $str, $substr]});
use Encode qw(encode decode);
$str = decode('utf-8', "これは、ペンです。");
$substr = decode('utf-8', "ペン");
say encode('utf-8', qq(index "$str", "$substr" -> @{[index $str, $substr]}));
実行結果:
index "This is a pen.", "pen" -> 10 
index "これは、ペンです。", "ペン" -> 12  
index "これは、ペンです。", "ペン" -> 4 
pen の前にある This is a  は空白も含めて合計で10文字なので、10 が表示されます。
解説
indexは、文字列 STR の中から、検索文字列 SUBSTR を探し、最初に見つかった位置を返します。検索文字が見つからない場合には、-1 が返ります。
省略可能な引数、POSITION には、検索開始位置を指定します（ディフォルトは0）。
POSITION を使うと部分文字列が２回め以降に出現する位置も確かめることが出来、部分文字列の長さに注意すれば部分文字列の出現回数を数えることなどが容易になります。
位置が 0 から始まることに留意しましょう。 0 は文字列の左端を表します。
位置は文字単位ではなくバイト数なので、ソースコードエンコーディングが UTF-8 で多バイト文字が交じると、文字数とバイト数に食い違いが生じます。
このような場合は Encode モジュールを使い内部形式( ''internal format'' )に変換します。
内部形式であれば、サロゲートペアにも対応できますが、合成文字は修飾コードと基底文字はそれぞれ１文字に数えられます。
[https://paiza.io/projects/cp_epUyDzcstlfbqp6Z82A?language=perl 別解（utf8プラグマを使う）]:
use v5.30.0;
use warnings;
use utf8;
my $str = "This is a pen.";
my $substr = "pen";
say qq(index "$str", "$substr" -> @{[index $str, $substr]});
use Encode qw(encode);
$str = "これは、ペンです。";
$substr = "ペン";
say encode('utf-8', qq(index "$str", "$substr" -> @{[index $str, $substr]}));
実行結果:
index "This is a pen.", "pen" -> 10  
index "これは、ペンです。", "ペン" -> 4 
utf8プラグマを使い、ソースコードエンコーディングが UTF-8 であることを明示すると、文字リテラルは内部形式に変換され index や length で処理されます。
この場合でも、出力するときに内部形式から UTF-8 にエンコードする必要があります。
[https://paiza.io/projects/cp_epUyDzcstlfbqp6Z82A?language=perl 別解（binmodeを使う）]:
use v5.30.0;
use warnings;
use utf8;
binmode STDOUT,":encoding(UTF-8)";
my $str = "This is a pen.";
my $substr = "pen";
say qq(index "$str", "$substr" -> @{[index $str, $substr]});
$str = "これは、ペンです。";
$substr = "ペン";
say qq(index "$str", "$substr" -> @{[index $str, $substr]});
実行結果:
index "This is a pen.", "pen" -> 10  
index "これは、ペンです。", "ペン" -> 4 
毎回エンコードせず、STDOUT のディフォルトエンコーディングを UTF-8 にかえました。
binmode STDIN,":encoding(UTF-8)";とbinmode STDERR,":encoding(UTF-8)";も同時に指定したほうがいいかもしれません。
テキストのエンコーディングは、Perlを使っていると度々トラブルのもとになるので、回避方法が幾つかある事を知っておくと、他人の書いたコードを読むときなどに役に立ちます。
ここで紹介した方法の他に、歌代さんのjcode.plなどもあるのですが、標準モジュールの範囲の説明に留めました。
書式:rindex (STR, SUBSTR, [POSITION])
機能:文字列STRの中で、部分文字列SUBSTRが最後に出現する位置を返します。
[https://paiza.io/projects/BcE2R3K_lHUDlhNTQWwY3Q?language=perl 組込み関数 rindex]:
use v5.30.0;
use warnings;
use utf8;
binmode STDOUT,":encoding(UTF-8)";
my $str = "I like pens and pencils.";
my $substr = "pen";
say qq(rindex "$str", "$substr" -> @{[rindex $str, $substr]});
$str = "私は筆と鉛筆が好きです。";
$substr = "筆";
say qq(rindex "$str", "$substr" -> @{[rindex $str, $substr]});
実行結果:
rindex "I like pens and pencils.", "pen" -> 16  
rindex "私は筆と鉛筆が好きです。", "筆" -> 5 
解説
rindexは、文字列 STR の中から、検索文字列 SUBSTR を探し、最後に見つかった位置を返します（「末尾からの位置を返す」との編集が過去にありましたが、間違いです）。検索文字が見つからない場合には、-1 が返ります。
省略可能な引数、POSITION には、検索開始位置を指定します（ディフォルトは0）。
書式:substr (EXPR, OFFSET, [LENGTH], [REPLACEMENT])
文字列 EXPR から、OFFSET 目以降のバイト列を返します。取り出す長さ LENGTH をバイト単位で指定できますが、省略した場合は文字列の最後まで取り出します。なお、utf8プラグマが有効な場合は、バイト単位ではなく文字単位で取り出すことができます。
位置情報 OFFSET は上述のとおり 0 から始まりますが、LENGTH は容量なので通常は 1 以上の値を指定します。
文字列 REPLACEMENT を指定すると、取り出される部分を REPLACEMENT で置換します。
[https://paiza.io/projects/BcE2R3K_lHUDlhNTQWwY3Q?language=perl 組込み関数 rindex]:
use v5.30.0;
use warnings;
use utf8;
binmode STDOUT,":encoding(UTF-8)";
my $str = "Hello, world!";
say substr($str, index($str, "world"), length("world"), "Japan");
say $str;
$str = "こんにちは、世界！";
say substr($str, index($str, "世界"), length("世界"), "🌍");
say $str;
実行結果:
world
Hello, Japan!
世界 
こんにちは、🌍！
書式:uc ([EXPR])
文字列 EXPR を大文字にして返します。EXPR を省略すると、$_ が使われます。
書式:ucfirst ([EXPR])
uc と同じですが、先頭1文字を大文字にして返します。
書式:lc ([EXPR])
uc と同じですが、小文字にして返します。
書式:lcfirst ([EXPR])
ucfirst と同じですが、小文字にして返します。
書式:
chop VARIABLE
chop (LIST)
変数 VARIABLE の末尾の末尾1文字を削除します。
変数のリストを渡された場合は、各変数について同じ処理を行います。
VARIABLE を省略すると $_ が使われます。
[https://paiza.io/projects/Fva3C8fhvsNLFSY7C4WNTw?language=perl chopとchomp]:
use v5.30.0;
use warnings;
use utf8;
binmode STDOUT,":encoding(UTF-8)";
my $str = "Hello, world!\n";
chop $str;
say "chop: $str(@{[length $str]})";
chop $str;
say "chop: $str(@{[length $str]})";
chop $str;
say "chop: $str(@{[length $str]})";
chop $str;
say "chop: $str(@{[length $str]})";
$str = "Hello, world!\n";
chomp $str;
say "chomp: $str(@{[length $str]})";
chomp $str;
say "chomp: $str(@{[length $str]})";
chomp $str;
say "chomp: $str(@{[length $str]})";
chomp $str;
say "chomp: $str(@{[length $str]})";
$str = "Hello, world!\n";
$str = substr($str, 0, length($str) - 1);
say "substr(): $str(@{[length $str]})";
$str = substr($str, 0, length($str) - 1);
say "substr(): $str(@{[length $str]})";
$str = substr($str, 0, length($str) - 1);
say "substr(): $str(@{[length $str]})";
sub chop(\$) {
    my $strr = shift;
    $$strr = substr($$strr, 0, length($$strr) - 1);
    undef
}
$str = "Hello, world!\n";
say "::chop: $str(@{[length $str]})";
say "::chop: $str(@{[length $str]})";
say "::chop: $str(@{[length $str]})";
sub chomp(\$) {
    my $strr = shift;
    $$strr = substr($$strr, 0, length($$strr) - 1) if substr($$strr, length($$strr) - 1, 1) eq "\n";
    undef
}
$str = "Hello, world!\n";
say "::chomp: $str(@{[length $str]})";
say "::chomp: $str(@{[length $str]})";
実行結果:
chop: Hello, world!(13)
chop: Hello, world(12)
chop: Hello, worl(11)
chop: Hello, wor(10)
chomp: Hello, world!(13)
chomp: Hello, world!(13)
chomp: Hello, world!(13)
chomp: Hello, world!(13)
substr(): Hello, world!(13)
substr(): Hello, world(12)
substr(): Hello, worl(11)
書式:
chomp VARIABLE
chomp (LIST)
変数 VARIABLE の末尾の $/（デフォルトは "\n"）を削除します。
変数のリストを渡された場合は、各変数について同じ処理を行います。
VARIABLE を省略すると $_ が使われます。
書式:chr [NUMBER]
文字セットで NUMBER 番目に割り当てられている文字を返します。
NUMBER を省略すると $_ が使われます。
逆の操作を行うには ord を使います。
書式:crypt PLAINTEXT, SALT
C ライブラリの crypt(3) をエミュレートします。
書式:hex [EXPR]
十六進数 EXPR を十進数に変換して返します。EXPR を省略すると $_ が使われます。
書式:length [EXPR]
文字列 EXPR の長さを返します。bytes プラグマが有効な場合（デフォルト）はバイト数を、utf8 プラグマが有効な場合は文字数を返します。EXPR を省略すると $_ が使われます。
書式:oct [EXPR]
八進数 EXPR を十進数に変換して返します。EXPR を省略すると $_ が使われます。
書式:ord [EXPR]
文字列 EXPR の文字セット上でのコード位置を返します。EXPR を省略すると $_ が使われます。逆の操作を行うには chr を使います。
書式:pack TEMPLATE, LIST
LIST を TEMPLATE に従ってバイナリデータに変換します。
q/STRING/
qq/STRING/
qr/STRING/
qx/STRING/
qw/STRING/
シングルクォート、ダブルクォート、正規表現、バッククォート、単語クォート。詳細は演算子の章を参照。
書式:reverse LIST
リストコンテキストでは LIST の順番を逆順にしたリストを返します。スカラーコンテキストでは LIST の要素を結合した後に逆順にした文字列を返します。
[https://paiza.io/projects/m274vBmTsDDTfyB1yKQy3g?language=perl 例]:
use v5.30.0;
use warnings;
my @array = qw(あい うえお かきくけこ 🏝);
say "@{[ reverse @array ]}";
say "@{[ scalar reverse @array ]}";
use utf8;
binmode STDOUT,":encoding(UTF-8)";
@array = qw(あい うえお かきくけこ 🏝);
say "@{[ reverse @array ]}";
say "@{[ scalar reverse @array ]}";
実行結果:
🏝 かきくけこ うえお あい
����㑁㏁㍁㋁㊁㈁ㆁ㄁め�
🏝 かきくけこ うえお あい 
🏝こけくきかおえういあ
Perlの文字列はディフォルトではバイトシーケンスなのでバイト逆順にすると多バイト文字は破綻し、上記のように文字化けします。
use utf8;で、バイトシーケンスから内部エンコーディング( ''Wide character'' )に切替えることができますが、このまま say すると内部エンコーディングのままなので、標準出力のレイヤーを ":encoding(UTF-8)" に変更します。
書式:sprintf FORMAT, LIST
LIST を FORMAT に従って整形して返します。
書式:tr///
1文字を対応する1文字に置換します。詳細は演算子の章を参照。
書式:y///
tr///と同義。
m//, pos, qr//, quotemeta, s///, split, study
abs, atan2, cos, exp, hex, int, log, oct, rand, sin, sqrt, srand
each, keys, pop, push, shift, splice, unshift, values
grep, join, map, qw//, reverse, sort, unpack
delete, each, exists, keys, values
binmode, close, closedir, dbmclose, dbmopen, die, eof, fileno, flock, format, getc, print, printf, read, readdir, readline, rewinddir, say, seek, seekdir, select, syscall, sysread, sysseek, syswrite, tell, telldir, truncate, warn, write
pack, read, syscall, sysread, sysseek, syswrite, unpack, vec
-X, chdir, chmod, chown, chroot, fcntl, glob, ioctl, link, lstat, mkdir, open, opendir, readlink, rename, rmdir, select, stat, symlink, sysopen, umask, unlink, utime
break, caller, continue, die, do, dump, eval, evalbytes, exit, __FILE__, goto, last, __LINE__, next, __PACKAGE__, redo, return, sub, __SUB__, wantarray
caller, import, local, my, our, package, state, use
defined, formline, lock, prototype, reset, scalar, undef
alarm, exec, fork, getpgrp, getppid, getpriority, kill, pipe, qx//, readpipe, setpgrp, setpriority, sleep, system, times, wait, waitpid
do, import, no, package, require, use
bless, dbmclose, dbmopen, package, ref, tie, tied, untie, use
accept, bind, connect, getpeername, getsockname, getsockopt, listen, recv, send, setsockopt, shutdown, socket, socketpair
msgctl, msgget, msgrcv, msgsnd, semctl, semget, semop, shmctl, shmget, shmread, shmwrite
endgrent, endhostent, endnetent, endpwent, getgrent, getgrgid, getgrnam, getlogin, getpwent, getpwnam, getpwuid, setgrent, setpwent
endprotoent, endservent, gethostbyaddr, gethostbyname, gethostent, getnetbyaddr, getnetbyname, getnetent, getprotobyname, getprotobynumber, getprotoent, getservbyname, getservbyport, getservent, sethostent, setnetent, setprotoent, setservent
gmtime, localtime, time, times
and, AUTOLOAD, BEGIN, catch, CHECK, cmp, CORE, __DATA__, default, defer, DESTROY, else, elseif, elsif, END, __END__, eq, finally, for, foreach, ge, given, gt, if, INIT, isa, le, lt, ne, not, or, try, UNITCHECK, unless, until, when, while, x, xor
each keys pop push shift splice unshift values
retrieve the next key/value pair from a hash
retrieve list of indices from a hash
remove the last element from an array and return it
append one or more elements to an array
remove the first element of an array, and return it
add or remove elements anywhere in an array
prepend more elements to the beginning of a list
return a list of the values in a hash
pack read syscall sysread sysseek syswrite unpack vec
convert a list into a binary representation
fixed-length buffered input from a filehandle
execute an arbitrary system call
fixed-length unbuffered input from a filehandle
position I/O pointer on handle used with sysread and syswrite
fixed-length unbuffered output to a filehandle
convert binary structure into normal perl variables
test or set particular bits in a string
-X chdir chmod chown chroot fcntl glob ioctl link lstat mkdir open opendir readlink rename rmdir select stat symlink sysopen umask unlink utime
a file test (-r, -x, etc)
change your current working directory
changes the permissions on a list of files
change the ownership on a list of files
make directory new root for path lookups
file control system call
expand filenames using wildcards
system-dependent device control system call
create a hard link in the filesystem
stat a symbolic link
create a directory
open a file, pipe, or descriptor
open a directory
determine where a symbolic link is pointing
change a filename
remove a directory
reset default output or do I/O multiplexing
get a file's status information
create a symbolic link to a file
open a file, pipe, or descriptor
set file creation mode mask
remove one link to a file
set a file's last access and modify times
break caller continue die do dump eval evalbytes exit __FILE__ goto last __LINE__ next __PACKAGE__ redo return sub __SUB__ wantarray
break out of a C block
get context of the current subroutine call
optional trailing block in a while or foreach
raise an exception or bail out
turn a BLOCK into a TERM
create an immediate core dump
catch exceptions or compile and run code
similar to string eval, but intend to parse a bytestream
terminate this program
the name of the current source file
create spaghetti code
exit a block prematurely
the current source line number
iterate a block prematurely
the current package
start this loop iteration over again
get out of a function early
declare a subroutine, possibly anonymously
the current subroutine, or C if not in a subroutine
get void vs scalar vs list context of current subroutine call
delete each exists keys values
deletes a value from a hash
retrieve the next key/value pair from a hash
test whether a hash key is present
retrieve list of indices from a hash
return a list of the values in a hash
binmode close closedir dbmclose dbmopen die eof fileno flock format getc print printf read readdir readline rewinddir say seek seekdir select syscall sysread sysseek syswrite tell telldir truncate warn write
prepare binary files for I/O
close file (or pipe or socket) handle
close directory handle
breaks binding on a tied dbm file
create binding on a tied dbm file
raise an exception or bail out
test a filehandle for its end
return file descriptor from filehandle
lock an entire file with an advisory lock
declare a picture format with use by the write() function
get the next character from the filehandle
output a list to a filehandle
output a formatted list to a filehandle
fixed-length buffered input from a filehandle
get a directory from a directory handle
fetch a record from a file
reset directory handle
output a list to a filehandle, appending a newline
reposition file pointer for random-access I/O
reposition directory pointer
reset default output or do I/O multiplexing
execute an arbitrary system call
fixed-length unbuffered input from a filehandle
position I/O pointer on handle used with sysread and syswrite
fixed-length unbuffered output to a filehandle
get current seekpointer on a filehandle
get current seekpointer on a directory handle
shorten a file
print debugging info
print a picture record
grep join map qw/STRING/ reverse sort unpack
locate elements in a list test true against a given criterion
join a list into a string using a separator
apply a change to a list to get back a new list with the changes
quote a list of words
flip a string or a list
sort a list of values
convert binary structure into normal perl variables
abs atan2 cos exp hex int log oct rand sin sqrt srand
absolute value function
arctangent of Y/X in the range -PI to PI
cosine function
raise I to a power
convert a hexadecimal string to a number
get the integer portion of a number
retrieve the natural logarithm for a number
convert a string to an octal number
retrieve the next pseudorandom number
return the sine of a number
square root function
seed the random number generator
defined formline lock prototype reset scalar undef
test whether a value, variable, or function is defined
internal function used for formats
get a thread lock on a variable, subroutine, or method
get the prototype (if any) of a subroutine
clear all variables of a given name
force a scalar context
remove a variable or function definition
do import no package require use
turn a BLOCK into a TERM
patch a module's namespace into your own
unimport some module symbols or semantics at compile time
declare a separate global namespace
load in external functions from a library at runtime
load in a module at compile time and import its namespace
caller import local my our package state use
get context of the current subroutine call
patch a module's namespace into your own
create a temporary value for a global variable (dynamic scoping)
declare and assign a local variable (lexical scoping)
declare and assign a package variable (lexical scoping)
declare a separate global namespace
declare and assign a persistent lexical variable
load in a module at compile time and import its namespace
endprotoent endservent gethostbyaddr gethostbyname gethostent getnetbyaddr getnetbyname getnetent getprotobyname getprotobynumber getprotoent getservbyname getservbyport getservent sethostent setnetent setprotoent setservent
be done using protocols file
be done using services file
get host record given its address
get host record given name
get next hosts record
get network record given its address
get networks record given name
get next networks record
get protocol record given name
get protocol record numeric protocol
get next protocols record
get services record given its name
get services record given numeric port
get next services record
prepare hosts file for use
prepare networks file for use
prepare protocols file for use
prepare services file for use
bless dbmclose dbmopen package ref tie tied untie use
create an object
breaks binding on a tied dbm file
create binding on a tied dbm file
declare a separate global namespace
find out the type of thing being referenced
bind a variable to an object class
get a reference to the object underlying a tied variable
break a tie binding to a variable
load in a module at compile time and import its namespace
alarm exec fork getpgrp getppid getpriority kill pipe qx/STRING/ readpipe setpgrp setpriority sleep system times wait waitpid
schedule a SIGALRM
abandon this program to run another
create a new process just like this one
get process group
get parent process ID
get current nice value
send a signal to a process or process group
open a pair of connected filehandles
backquote quote a string
execute a system command and collect standard output
set the process group of a process
set a process's nice value
block for some number of seconds
run a separate program
return elapsed time for self and child processes
wait for any child process to die
wait for a particular child process to die
m// pos qr/STRING/ quotemeta s/// split study
match a string with a regular expression pattern
find or set the offset for the last/next m//g search
compile pattern
quote regular expression magic characters
replace a pattern with a string
split up a string using a regexp delimiter
no-op, formerly optimized input data for repeated searches
accept bind connect getpeername getsockname getsockopt listen recv send setsockopt shutdown socket socketpair
accept an incoming socket connect
binds an address to a socket
connect to a remote socket
find the other end of a socket connection
retrieve the sockaddr for a given socket
get socket options on a given socket
register your socket as a server
receive a message over a Socket
send a message over a socket
set some socket options
close down just half of a socket connection
create a socket
create a pair of sockets
chomp chop chr crypt fc hex index lc lcfirst length oct ord pack q/STRING/ qq/STRING/ reverse rindex sprintf substr tr/// uc ucfirst y///
remove a trailing record separator from a string
remove the last character from a string
get character this number represents
one-way passwd-style encryption
return casefolded version of a string
convert a hexadecimal string to a number
find a substring within a string
return lower-case version of a string
return a string with just the next letter in lower case
return the number of characters in a string
convert a string to an octal number
find a character's numeric representation
convert a list into a binary representation
singly quote a string
doubly quote a string
flip a string or a list
right-to-left substring search
formatted print into a string
get or alter a portion of a string
transliterate a string
return upper-case version of a string
return a string with just the next letter in upper case
transliterate a string
msgctl msgget msgrcv msgsnd semctl semget semop shmctl shmget shmread shmwrite
SysV IPC message control operations
get SysV IPC message queue
receive a SysV IPC message from a message queue
send a SysV IPC message to a message queue
SysV semaphore control operations
get set of SysV semaphores
SysV semaphore operations
SysV shared memory operations
get SysV shared memory segment identifier
read SysV shared memory
write SysV shared memory
gmtime localtime time times
convert UNIX time into record or string using Greenwich time
convert UNIX time into record or string using local time
return number of seconds since 1970
return elapsed time for self and child processes
endgrent endhostent endnetent endpwent getgrent getgrgid getgrnam getlogin getpwent getpwnam getpwuid setgrent setpwent
be done using group file
be done using hosts file
be done using networks file
be done using passwd file
get next group record
get group record given group user ID
get group record given group name
return who logged in at this tty
get next passwd record
get passwd record given user login name
get passwd record given user ID
prepare group file for use
prepare passwd file for use